

HEPMath 1.1: A Mathematica Package for Semi-Automatic Computations in High Energy Physics

MARTIN WIEBUSCH*

Institute for Particle Physics Phenomenology, Department of Physics, Durham University, Durham CH1 3LE, United Kingdom

Abstract

This article introduces the Mathematica package *HEPMath* which provides a number of utilities and algorithms for High Energy Physics computations in Mathematica. Its functionality is similar to packages like FormCalc or FeynCalc, but it takes a more complete and extensible approach to implementing common High Energy Physics notations in the Mathematica language, in particular those related to tensors and index contractions. It also provides a more flexible method for the generation of numerical code which is based on new features for C code generation in Mathematica. In particular it can automatically generate *Python extension modules* which make the compiled functions callable from Python, thus eliminating the need to write any code in a low-level language like C or Fortran. It also contains seamless interfaces to LHAPDF, FeynArts, and LoopTools.

*email: martin.wiebusch@durham.ac.uk

Contents

1	Introduction	2
2	Installation	4
3	Building Blocks for Feynman Amplitudes	7
3.1	Tensors and Dirac-Algebra	7
3.2	Lorentz Vectors and Dirac Spinors	11
3.3	Conjugation	14
3.4	Colour Algebra	14
4	Algebraic Manipulations	15
4.1	Expanding and Collecting	16
4.2	Index Manipulations and Substitutions	19
4.3	Tensor Derivatives	21
4.4	Simplification of Colour Structures	21
4.5	Polarisation Sums	22
4.6	One-Loop Integrals	23
4.7	Feynman Parametrisation of Multi-Loop Integrals	25
5	User-Defined Tensors and Linear Functions	26
6	Interfaces	28
6.1	FeynArts	28
6.2	LoopTools	31
6.3	LHAPDF	31
7	Code Generation	32
8	Conclusions	36

1 Introduction

In many areas of High Energy Physics the scientific progress is intrinsically linked to progress in the development of computational tools and the implementation of computational algorithms. A prime example for this interplay is the numerical computation and integration of scattering cross sections. For tree-level cross sections in the Standard Model (SM) automatisation has been accomplished about 10 years ago [1–12]. The implementation of virtually arbitrary models has been automated in [13–19]. More recently substantial progress has been made towards the automatisation of the computation next-to-leading order cross sections [20–23]. These tools aim to be *fully* automatic in the sense that they implement a fixed computational scheme and the only user input required is information about the process, model parameters, perturbative order etc.

There are, however, other types of calculations in high energy physics where full automatisation (in the above-mentioned sense) is either not possible or not yet available. Such cases include the computation of Wilson coefficients or anomalous dimensions in effective field theories and calculations where the result is needed in analytic form or where process-specific manipulations (expansions in external momenta etc.) are required at some stage in the symbolic computation. These computations can still be greatly simplified by automatisation, but they call for a different type of tool, which I will refer to as *semi-automatic* tools. Semi-automatic tools try to mirror conventional mathematical notations in a programming environment and provide a common interface to as many standard algorithms as possible (e.g. the computation of Dirac traces or the reduction of loop integrals). To offer the necessary flexibility they must, however, leave control over the computational scheme in the hands of the user. The *HEPMath* package presented in this paper falls in this category. It provides the following functionality:

1. Consistent integration of tensors with symbolic indices and index contractions in Mathematica, along with functions to manipulate (expand, collect etc.) such expressions. Arbitrary tensors and index types can be defined by the user.
2. Implementation of the basic building blocks of Feynman amplitudes, such as Dirac spinors, Dirac matrices, Lorentz vectors, colour matrices etc. and their algebraic properties.
3. Computation of traces over Dirac matrices in 4 and D dimensions.
4. Simplification of colour structures.
5. Passarino-Veltman reduction of one-loop integrals. Currently only the reduction to tensor integrals is implemented, but this is likely to change in the future.
6. Seamless interfaces to FeynArts [24, 25], LoopTools [2] and LHAPDF [26].
7. Generation and compilation of numerical code, based on Mathematica's `Compile` and `CCodeGenerate` functions.

8. Generation of *Python extension modules* which make the generated functions callable from Python.

Feature 8 is new to the best of my knowledge, but there are of course several other publicly available semi-automatic tools which provide a subset of the above-mentioned functionality. Some also provide functionality that is not present in HEPMath (yet). In the following I will briefly review these tools and highlight the differences and similarities with HEPMath.

FORM [27] is clearly one of the most established tools for symbolic computations in High Energy Physics. Its functionality covers items 1 to 3 and it is highly optimised for handling large polynomial expressions. It is a standalone program with its own input language which, despite its flexibility, can not be regarded as a general purpose programming language. The design of FORM puts performance before ease of use, while HEPMath is mainly intended as a convenience tool. Simply put, FORM will reduce the *run time* of your symbolic computations while HEPMath tries to reduce the *development time*. In many applications the latter vastly outweighs the former, and HEPMath is written for these cases.

TRACER [28] is a Mathematica package which calculates traces over Dirac matrices. Its functionality covers item 3 and some aspects of item 2. It can be regarded as an ancestor of HEPMath, since the HEPMath algorithm for the computation of Dirac traces was taken from TRACER. However, the TRACER syntax for constructing tensor expressions (such as contractions between Lorentz vectors etc.) is ambiguous¹ and was completely re-designed in HEPMath.

FormCalc [2, 29–34] is a Mathematica package for the calculation of tree-level and one-loop amplitudes. It is developed in parallel with FeynArts [24, 25] and LoopTools [2] and its functionality covers items 2 to 7. It uses FORM as a back-end for parts of the symbolic computation and thus offers the performance of FORM through a user friendly Mathematica interface. Its approach differs from HEPMath in the sense that its computational scheme, although configurable with many options, is ultimately pre-defined and out of the user’s control. Also, the structure of numerical code generated with FormCalc is mostly fixed while the code generation mechanism of HEPMath offers a lot more freedom.

FeynCalc [35] is a Mathematica package whose goals and design principles are very similar to those of HEPMath. Its functionality includes items 2 to 6. It also provides several symbolic algorithms which are not present in HEPMath. In addition to features 1, 7, and 8 an advantage of HEPMath over FeynCalc is the absence of an ‘internal’ representation of expressions involving tensors (i.e. Lorentz vectors,

¹Try to make a substitution like $p \rightarrow xk$ in TRACER, where p and k are Lorentz vectors and x a scalar. There is no way to tell TRACER that x is the scalar and k the vector.

Dirac matrices etc.), which simplifies the implementation of additional symbolic algorithms by the user.

For any but the most trivial calculations it is generally unwise to rely on a single tool for automatic computations. With the exception of TRACER, HEPMath has no code overlap with any of the above-mentioned tools. It can therefore be used as an independent check of automatic calculations performed with any of them (except TRACER). The goal of the HEPMath project is to surpass these tools not so much in speed but in flexibility and ease-of-use. But of course such criteria are somewhat subjective.

This article is a user's guide for the HEPMath package. It documents its main functionality and illustrates its use with many examples.

2 Installation

The HEPMath package was developed with Mathematica 9 on a 64bit Linux platform. I will only describe the installation process on Linux here. The latest version of the HEPMath package is available from

<http://hepmath.hepforge.org>

After downloading the tarball `hepmath-x.y.tar.gz` (where 'x.y' stands for the current version number) run the following commands in a terminal:

```
tar -zxvf hepmath-x.y.tar.gz
cd hepmath-x.y
./configure [options]
make
make install
```

All necessary information for the installation can be specified through command line options to the `configure` script. The order of the arguments is irrelevant.

As for all GNU packages, the installation prefix for various libraries and Python modules included in HEPMath can be set with the `--prefix` option. The default setting typically requires administrator rights, and if you don't have them you might want to change it, e.g. to

```
--prefix=my-prefix
```

If you want to generate Python extension modules with HEPMath you may want to use a prefix that your Python interpreter searches for Python packages. On my system this is `$HOME/.local`.

To install only the Mathematica packages without the LHAPDF and LoopTools interfaces no additional options need to be passed to `configure`. Note that `make` needs to execute a few Mathematica scripts. It assumes that the command which starts the command line version of Mathematica is called `math`. If the executable has a different name, say `math9`, on your system (e.g. because you have several Mathematica versions installed) you can tell `configure` about this by passing

```
MATH=math9
```

as a command line argument. The Mathematica modules will be installed in `$HOME/.Mathematica/Applications/`, independent of your `--prefix` setting. You can change the installation directory of the Mathematica modules to `path` with

```
MATHDIR=path
```

However, it is your responsibility to assure that Mathematica searches `path` for packages, e.g. by setting the `$UserBaseDirectory` variable in Mathematica accordingly.

HEPMath can access parton distribution functions by interfacing with the LHAPDF library. This obviously requires a working installation of the LHAPDF package. In particular you must make sure that the library `libLHAPDF.so` from by the LHAPDF package is installed in a directory where your dynamic linker can find it. This can, for example, be accomplished by adding the directory containing `libLHAPDF.so` to your `LD_LIBRARY_PATH` environment variable. The HEPMath interface to LHAPDF was tested with LHAPDF version 6.1.3. To enable it use one of the following options:

```
--with-lhapdf  
--with-lhapdf=path
```

Use the first form if you have installed LHAPDF under a standard prefix (e.g. `/usr` or `/usr/local`) or if you have configured your system so that your C compiler and linker find the LHAPDF headers and libraries without additional flags. The second form allows you to specify the prefix `path` under which LHAPDF is installed. Alternatively, you can specify the compiler and linker flags required for LHAPDF with

```
LHAPDF_CFLAGS=compile-flags  
LHAPDF_LIBS=link-flags
```

One-loop tensor integrals can be evaluated numerically in HEPMath through an interface to the LoopTools package. The LoopTools interface was tested with LoopTools version 2.9. To install it you will most likely need to re-compile the LoopTools library. The reason is that LoopTools creates a static library by default, while the HEPMath interface requires a dynamic library. Converting a static to a dynamic library is only possible when the static library was compiled with the right flags, and that's why you need to re-compile. On most Linux platforms this can be done as follows:

```
cd looptools-path
make clean
CFLAGS=-fPIC FFLAGS=-fPIC defs ./configure opts
make
```

where *looptools-path* is the path to your LoopTools source directory and *defs* and *opts* are any additional variable definitions or options you like to use when compiling LoopTools. After successfully re-compiling LoopTools pass the option

```
--with-looptools=looptools-path
```

to the HEPMath `configure` script. Since LoopTools is written in Fortran and the HEPMath interface in C the correct Fortran run-time libraries are needed to link the interface. Usually `configure` will find them automatically, but if you have several Fortran compilers installed on your system it might pick the wrong one. (It needs the one you used to compile LoopTools, but it has no way of detecting which one that was.) You can point it to the right compiler, say, `gfortran` with the option

```
F77=gfortran
```

Alternatively, you can specify the options needed to compile and link C code with LoopTools with the following options

```
LOOPTOOLS=/path/to/libooptools.a
LOOPTOOLS_CFLAGS=compile-flags
LOOPTOOLS_LIBS=link-flags
```

where *link-flags* should *not* include the `-looptools` flag.

Note that HEPMath does not use the the Mathematica interface provided in the LoopTools distribution, and you do not need to compile this interface in order to access LoopTools from HEPMath. The reason for this duplication of interfaces is that the one provided by LoopTools is based on MathLink. The code generation method used in HEPMath can only handle calls to external functions if these functions are interfaced via LibraryLink, and therefore HEPMath must implement its own interface to LoopTools.

Once you have HEPMath installed correctly you can try and run the examples in the `examples` sub-directory of the source directory. The provided examples are:

ee-mumu.m The computation of the $e^+e^- \rightarrow \mu^+\mu^-$ squared matrix element in QED. This example does not require FeynArts, LHAPDF or LoopTools.

hepcompile.m A short demonstration of the code generation mechanism, including the generation of a Python extension module. This example requires the LoopTools interface (but you can easily modify it so that it doesn't).

H-gg-SM.m The computation of the top-loop-induced $H \rightarrow gg$ decay width in the SM at leading order. This example illustrates the use of the FeynArts interface, the computation of one-loop integrals with LoopTools, colour algebra and code generation. It requires FeynArts and the LoopTools interface.

I recommend to run the files as scripts, i.e. with

```
math -script example.m
```

The source code in the examples is well commented and should get you started quickly with HEPMath. For a more structured introduction just continue reading.

3 Building Blocks for Feynman Amplitudes

If HEPMath is properly installed you can open a Mathematica session and load the package with

```
Needs["HEPMath`"]
```

The HEPMath package defines a number of symbols which represent typical building blocks of Feynman amplitudes within Mathematica. This section summarises these building blocks and describes their pre-defined behaviour.

3.1 Tensors and Dirac-Algebra

Dirac matrices are represented by the symbol `Ga`. Try evaluating

```
HEPTensorSignature[Ga]
⇒ {Lorentz, Dirac, Dirac}
```

(The text after the arrow '⇒' is the expression returned by Mathematica.) This tells you that `Ga` is an object which carries three indices: one Lorentz index and two Dirac indices. All objects which can carry indices are called *HEPTensors* in HEPMath, and the `HEPTensorSignature` function gives you a list with the types of indices you can attach. In particular, any variable not known to HEPMath is a *scalar*, i.e. has an empty tensor signature.

```
HEPTensorSignature[xyz]
⇒ {}
```

To attach symbolic indices to `Ga` simply apply it to the list of indices.

```
HEPTensorSignature[Ga[mu, alpha, beta]]
⇒ {}
HEPTensorSignature[Ga[mu]]
⇒ {Dirac, Dirac}
```

Attaching the full list of indices turns `Ga` into a scalar, i.e. into an object with empty tensor signature. By attaching an incomplete list of indices you reduce the rank of the tensor, removing elements from its tensor signature from the left. The full set of pre-defined tensors provided by `HEPMath` is summarised in Tab. 1.

You can form linear combinations of tensors with the same tensor signature and use them in the same way as you would use a tensor symbol. For example

```
HEPTensorSignature[a GI + b G5]
⇒ {Dirac, Dirac}
HEPTensorSignature[(a GI + b G5)[al, bt]]
⇒ {}
```

(See Tab. 1 for an explanation of `GI` and `G5`.) If you try to add two tensors with different tensor signature or multiply two tensors `HEPMath` will complain.

```
HEPTensorSignature[a Ga + b G5]
⇒ HEPTensorSignature::conflict:
   Incompatible tensors in tensor expression b G5 + a Ga.
⇒ $Aborted
HEPTensorSignature[GI G5]
⇒ HEPTensorSignature::conflict:
   Incompatible tensors in tensor expression GI G5.
⇒ $Aborted
```

Implicit summation over repeated indices is understood and fully supported by `HEPMath`. You can get a summary of the free and contracted indices in an expression with the `Indices` function.

```
Indices[Ga[mu, al, bt] Ga[mu, bt, ga]]
⇒ {{al -> Dirac, ga -> Dirac}, {bt -> Dirac, mu -> Lorentz}}
```

Table 1: Pre-defined tensors and tensor-valued functions.

Symbol	Tensor signature	Description
Ga	{Lorentz, Dirac, Dirac}	The Dirac matrix γ^μ .
GI	{Dirac, Dirac}	The identity matrix $\mathbb{1}$.
G5	{Dirac, Dirac}	The matrix γ_5 .
PL	{Dirac, Dirac}	$(\mathbb{1} - \gamma_5)/2$.
PR	{Dirac, Dirac}	$(\mathbb{1} + \gamma_5)/2$.
Gs[p] ^a	{Dirac, Dirac}	The contraction $\gamma^\mu p_\mu \equiv \not{p}$.
GSig	{Lorentz, Lorentz, Dirac, Dirac}	The matrix $\sigma^{\mu\nu} = \frac{i}{2}(\gamma^\mu \gamma^\nu - \gamma^\nu \gamma^\mu)$.
USp[p] ^a	{FermionHelicity, Dirac}	The Dirac spinor $u(p)$.
VSp[p] ^a	{FermionHelicity, Dirac}	The Dirac spinor $v(p)$.
Pol[p] ^a	{VectorPolarization, Lorentz}	The massless polarisation vector $\varepsilon^\mu(p)$.
MPol[p] ^a	{MassiveVectorPolarization, Lorentz}	The massive polarisation vector $\varepsilon^\mu(p)$.
Dim	{}	Symbol representing the dimension of Minkowski space.
Eta	{Lorentz, Lorentz}	The metric tensor $\eta^{\mu\nu} \equiv g^{\mu\nu}$ in Dim dimensions with “mostly minus” signature.
EtaHat	{Lorentz, Lorentz}	The (Dim - 4)-dimensional part of the metric tensor.
Eps	{Lorentz, Lorentz, Lorentz, Lorentz}	The Levi-Civita tensor $\varepsilon^{\mu\nu\rho\sigma}$.
Den[p, m] ^a	{}	The propagator denominator $1/(p^2 - m^2)$.
ColorT	{ColorAdjoint, ColorFundamental, ColorFundamental}	The $SU(3)$ generator T^a .
ColorF	{ColorAdjoint, ColorAdjoint, ColorAdjoint}	The $SU(3)$ structure constants f^{abc} .
ColorDelta	{ColorFundamental, ColorFundamental}	The identity matrix for colour indices.
GluonDelta	{ColorAdjoint, ColorAdjoint}	The identity matrix for adjoint colour indices.

^a p must be an expression with tensor signature {Lorentz}.

The first list holds the free indices of the expression and the second list the contracted indices. Note how the types of the indices are inferred from their position in the index list. You can also use the `Indices` function to find out if an expression is well-formed:

```
Indices[Ga[mu, a1, a1] Ga[mu, a1, a1]]
⇒ Indices::mult: Index repeated more than twice in product
   Ga[mu, a1, a1] Ga[nu, a1, a1].
⇒ $Aborted
```

HEPMath provides a few index-free notations for contractions between tensors and functions for switching between indexed and index-free notation. Try evaluating the following expressions:

```
DiracContract[Ga[mu, a1, bt] Ga[nu, bt, ga]]
⇒ ( Ga[mu] . Ga[nu] )[a1, ga]
DiracContract[Ga[mu, a1, bt] Ga[nu, bt, a1]]
⇒ DiracTr[Ga[mu] . Ga[nu]]
```

As you see, Mathematica's dot operator (`Dot`) can be used to contract Dirac matrices. More generally, it will contract tensors with at most two indices of the same type in the obvious way. The expression `Ga[mu].Ga[nu]` has tensor signature `{Dirac, Dirac}` and can therefore be indexed with two Dirac indices. (Feel free to verify this with `HEPTensorSignature`.) `DiracTr` represents the trace over Dirac indices and can be applied to any object with tensor signature `{Dirac, Dirac}`. If you prefer to have all indices written out explicitly you can undo the effect of `DiracContract` with

```
ExplicitIndices[(Ga[mu] . Ga[nu])[a1, ga]]
⇒ Ga[mu, a1, $3] Ga[nu, $3, ga]
```

Note how HEPMath generates unique symbols for the dummy indices.

The Dirac trace can be calculated with the `CalcDiracTraces` function.

```
CalcDiracTraces[DiracTr[Ga[mu] . Ga[nu]]]
⇒ 4 Eta[mu, nu]
```

The symbol `Eta` represents the metric tensor $\eta^{\mu\nu} \equiv g^{\mu\nu}$ in $D \equiv \text{Dim}$ dimensions with the “mostly minus” signature. The `CalcDiracTraces` function computes any occurrence of `DiracTr[...]` in its argument and substitutes the result. Occurrences of γ_5 matrices (represented by `G5`, see Tab. 1) are handled the same way as in TRACER, i.e. by letting the $(D-4)$ -dimensional part of γ^μ commute with γ_5 . The option `AntiCommutingG5` lets you use a completely anti-commuting γ_5 instead:

```

CalcDiracTraces[DiracTr[Ga[mu].G5.Ga[nu].G5]]
⇒ -4 Eta[mu, nu] + 8 EtaHat[mu, nu]
CalcDiracTraces[DiracTr[Ga[mu].G5.Ga[nu].G5], AntiCommutingG5->True]
⇒ -4 Eta[mu, nu]

```

The symbol `EtaHat` represents the $(D - 4)$ -dimensional part of the metric tensor. You can check the dimension of `Eta` and `EtaHat` with the `LorentzContract` function:

```

LorentzContract[Eta[mu,mu]]
⇒ Dim
LorentzContract[EtaHat[mu,mu]]
⇒ -4 + Dim

```

The Levi-Civita tensor is represented by the symbol `Eps`. Its indices live in 4 dimensions.

```

LorentzContract[Eps[mu, nu, ro, sg] EtaHat[sg, lm]]
⇒ 0

```

You can use the `SortEps` function to bring the arguments of `Eps` in canonical order (as defined by Mathematica's `OrderedQ` function)

```

SortEps[a Eps[nu, mu, ro, sg] + b Eps[mu, nu, ro, nu]]
⇒ -a Eps[mu, nu, ro, sg]

```

3.2 Lorentz Vectors and Dirac Spinors

Playing with the pre-defined tensors of `HEPMath` is nice, but for an actual calculation you also need to be able to define your own tensors. At the very least you will need symbols that represent Lorentz vectors. To tell `HEPMath` that a certain symbol represents a Lorentz vector you have to declare it.

```

DeclareLorentzVectors[p]
DeclareLorentzVectors[p, k, q]
DeclareLorentzVectors[{p, k, q}]

```

These commands tell `HEPMath` that the symbols `p`, `k`, and `q` represent *real*, D -dimensional Lorentz vectors. The arguments of `DeclareLorentzVectors` can be arbitrary *patterns*. Any expression matching one of these patterns will be recognised as a Lorentz vector. For example

```

DeclareLorentzVectors[_p]

```

makes any expression of the form $p[\mathit{something}]$ a Lorentz vector.

Complex and/or four-dimensional Lorentz vectors can be declared with the functions

```
Declare4DLorentzVectors
DeclareComplexLorentzVectors
DeclareComplex4DLorentzVectors
```

Similar to `DiracContract` the `LorentzContract` function can be used to eliminate unnecessary occurrences of `Eta` and to write contractions between Lorentz vectors in index-free form.

```
LorentzContract[Eta[mu, nu] p[nu]]
⇒ p[mu]
LorentzContract[k[mu] p[mu]]
⇒ SDot[k, p]
```

The last expression is equivalent to $k \cdot p$ but holds the additional information that the product is symmetric. (The `SDot` symbol has the attribute `Orderless`.) Although mostly used for products of Lorentz vectors, `SDot` is more general and can represent the full contraction of any two tensors with the same tensor signature.

```
ExplicitIndices[SDot[Eta, EtaHat]]
⇒ Eta[$3, $4] EtaHat[$3, $4]
```

The full contraction of a tensor with itself is represented by the `Sqr` function:

```
SDot[p, p]
⇒ Sqr[p]
ExplicitIndices[Sqr[p]]
⇒ p[$3]^2
```

Note that `HEPMath` is aware that the index `$3` in $p[\$3]^2$ is contracted.

```
Indices[p[$3]^2]
⇒ {}, {$3 -> Lorentz}}
```

The distinguishing feature of four-dimensional Lorentz vectors is that their contractions with `EtaHat` vanish:

```
Declare4DLorentzVectors[p];
LorentzContract[EtaHat[mu, nu] p[nu]]
⇒ 0
```

HEPMath also defines a few “tensor-valued functions”, i.e. symbols which become a tensor only when they are applied to something else. For example, Dirac spinors $u(p)$ and $v(p)$ associated with a four-momentum p can be represented by `USp[p]` and `VSp[p]`, respectively.

```
DeclareLorentzVectors[p];
HEPTensorSignature[USp[p]]
⇒ {FermionHelicity, Dirac}
HEPTensorSignature[VSp[p]]
⇒ {FermionHelicity, Dirac}
```

Note that the helicities of the spinors are represented by the first index whose type is `FermionHelicity`. Polarisation vectors of massless or massive vector bosons can be represented by `Pol` and `MPol` in an analogous way.

```
DeclareLorentzVectors[p];
HEPTensorSignature[Pol[p]]
⇒ {VectorPolarization, Lorentz}
HEPTensorSignature[MPol[p]]
⇒ {MassiveVectorPolarization, Lorentz}
```

Contractions of Lorentz vectors with the Levi-Civita tensor are represented by `EpsDot`. It takes up to four Lorentz vectors as arguments and its tensor signature depends on the number of supplied arguments.

```
DeclareLorentzVectors[k,l,p,q];
LorentzContract[Eps[mu, nu, ro, sg] k[mu] l[nu] p[ro] q[sg]]
⇒ EpsDot[k, l, p, q]
LorentzContract[Eps[mu, nu, ro, sg] k[mu] l[nu] p[ro]]
⇒ EpsDot[k, l, p][sg]
LorentzContract[Eps[mu, nu, ro, sg] k[mu] l[nu]]
⇒ EpsDot[k, l][ro, sg]
LorentzContract[Eps[mu, nu, ro, sg] k[mu]]
⇒ EpsDot[k][mu, ro, sg]
```

As for `Eps`, you can use the `SortEps` function to bring `EpsDot` expressions in canonical order.

```
SortEps[a EpsDot[p, q][nu, mu] + b EpsDot[q, p][mu, nu] +
        c EpsDot[p, p][mu, nu] + d EpsDot[p, q][mu, mu]]
⇒ -a*EpsDot[p, q][mu, nu] - b*EpsDot[p, q][mu, nu]
SortEps[EpsDot[l, k, p, q]]
⇒ -EpsDot[k, l, p, q]
```

3.3 Conjugation

Complex conjugation is represented in HEPMath by the function `CC`. Unlike Mathematica's `Conjugate` function it will automatically distribute itself over sums and products and it will automatically disappear when acting on a symbol that has been declared as real. To declare a symbol as real you can use the `DeclareReal` function.

```
DeclareLorentzVectors[p];
DeclareComplexLorentzVectors[e];
DeclareReal[r];
CC[c p[mu] + r e[mu]]
⇒ CC[c] p[mu] + r CC[e][mu]
```

Note that in the second term `CC` is applied to the vector `e`, not the indexed vector `e[mu]`.

The 'bar' notation for Dirac spinors and matrices is represented by the function `Bar`. Like `CC` it distributes itself over sums and products. The relationship between complex conjugation and bar conjugation is fully implemented:

```
Bar[r USp[p][s] + c VSp[p][s]]
⇒ r Bar[USp[p]][s] + CC[c] Bar[VSp[p]][s]
CC[Bar[USp[p]][s].PL.Gs[p].PR.Gs[e].G5.VSp[p][s]]
⇒ Bar[VSp[p]][s].(-G5).Gs[CC[e]].PL.Gs[p].PR.USp[p][s]
```

Note that I attached an explicit spin index `s` to the Dirac spinors above, but suppressed the Dirac indices.

3.4 Colour Algebra

$SU(3)$ generators T^a and structure constants f^{abc} are represented in HEPMath by the `ColorT` and `ColorF` tensors, respectively. The index types for the fundamental and adjoint $SU(3)$ representation are called `ColorFundamental` and `ColorAdjoint`. The identity matrices in these two index spaces are called `ColorDelta` and `GluonDelta`, respectively. There is no equivalent for `DiracContract` or `LorentzContract`, but you can remove unnecessary appearances of identity matrices (in any index space) with the `HEPContract` function.

```
HEPContract[ColorDelta[i, j] ColorT[a, j, k]]
⇒ T[a, i, k]
HEPContract[GluonDelta[a, b] ColorF[b, c, d]]
⇒ ColorF[a, c, d]
```

Traces over fundamental colour indices are represented by the `ColorTr` function, and the hermicity of the $SU(3)$ generators is implemented consistently.

```
CC[ColorT[a, i, j]]
⇒ T[a, j, i]
CC[ColorTr[ColorT[a].ColorT[b].ColorT[c]]]
⇒ ColorTr[ColorT[c] . ColorT[b] . ColorT[a]]
```

4 Algebraic Manipulations

In the last section I described the building blocks which HEPMath provides in order to build expressions representing Feynman amplitudes. Now it is time to learn how to put them together and manipulate them. To do this you first need to know which kind of constructs are “understood” by the HEPMath system. HEPMath follows a design principle which I call *substitution invariance* and which can be summarised by the following statement: *substituting valid expressions by other valid expressions with the same tensor signature in valid expressions must yield valid expressions*. What this means is simple. Take the expression `p[mu]` which represents a Lorentz vector \mathbf{p} with an index `mu` attached to it. When we replace `p` (a valid expression with tensor signature `{Lorentz}`) by the sum `k+q` of two other vectors (also a valid expression with tensor signature `{Lorentz}`) we get `(k+q)[mu]`. So, by substitution invariance `(k+q)[mu]` must be a valid expression if `p[mu]` is valid. The importance of substitution invariance is evident when you imagine some complicated expression in which you want to substitute `p` by `k+q`. The vector `p` can appear in various forms: `p[mu]`, `Sqr[p]`, `Gs[p]`, `USp[p][s]` etc. Writing a separate replacement rule for each of these cases is very error-prone and likely to make your code extremely hard to read. Therefore, HEPMath must be able to handle whatever expressions arise from the single substitution `p -> k+q`.

The main consequence of substitution invariance is that the following “funny” expressions must be valid:

- composite heads: `(a k + b q)[mu]`,
- multiple index lists: `Ga[mu][a1, bt]`.

Standard Mathematica functions for algebraic manipulations such as `Expand` or `Collect` obviously don’t know what to do with these expressions. HEPMath therefore complements Mathematica’s library of symbolic manipulation functions with functions that can handle tensor expressions, including the “funny” ones above. In addition it implements several “HEP-specific” functions which are frequently needed in High Energy Physics computations. This section describes both types of functions, starting with the basic ones. Some of the functions have already been mentioned in Sec. 3, but I list them here again for completeness.

4.1 Expanding and Collecting

HEPExpand mimics Mathematica's `Expand` function. Like `Expand` it takes a pattern as an optional second argument. All sub-expressions which are free of this pattern are not expanded. However, unlike `Expand`, **HEPExpand**'s default behaviour is to leave sub-expressions alone which do not contain any `HEPTensors`.

```
DeclareLorentzVectors[p, q];
HEPExpand[((a + b) (p + q)) [mu]]
⇒ (a + b) p[mu] + (a + b) q[mu]
HEPExpand[(a (p + q)) [mu], p]
⇒ a p[mu] + (a q) [mu]
```

Note how in the second case the factor `a` has not been pulled out of the head of the expression `(a q) [mu]` since this expression does not contain `p`. Needless to say, **HEPExpand** operates correctly on traces and index-free contractions such as `Gs`, `Sqr`, `SDot` and `Dot`.

```
HEPExpand[Bar[USp[p]] [s] . (Gs[p + q] + m GI) . VSp[q] [s]]
⇒ m Bar[USp[p]] [s] . GI . VSp[q] [s] +
   Bar[USp[p]] [s] . Gs[p] . VSp[q] [s] +
   Bar[USp[p]] [s] . Gs[q] . VSp[q] [s]
HEPExpand[DiracTr[Ga[mu] . (a GI + b G5) . Ga[nu]]]
⇒ b DiracTr[Ga[mu] . G5 . Ga[nu]] + a DiracTr[Ga[mu] . GI . Ga[nu]]
HEPExpand[Sqr[p + q]]
⇒ 2 SDot[p, q] + Sqr[p] + Sqr[q]
```

PullScalars Sometimes you just want to pull scalar factors from the heads of indexed expressions or linear functions like `SDot` without expanding everything. `HEPMath` gives you the function `PullScalars` to do that.

```
PullScalars[(a (p + q)) [mu]]
⇒ a (p + q) [mu]
PullScalars[DiracTr[Gs[a (p + q)] . Gs[Sqr[p] q]], _Sqr]
⇒ DiracTr[Gs[a (p + q)] . Gs[q]] Sqr[p]
```

The optional second argument can be any Mathematica pattern. Only scalars matching the pattern are pulled out.

SortEps was already mentioned in Sec. 3.2. It uses anti-symmetry of the Levi-Civita tensor to put the arguments of `Eps` and `EpsDot` expressions in canonical order.

HEPCollect Factorising expressions in `HEPMath` is complicated by the fact that tensor expressions can be put together with more operators than addition and multiplication. Just remember the different forms in which a Lorentz vector can appear: `p[mu]`, `Sqr[p]`, `SDot[p, q]` etc. The **HEPCollect** function implements

a behaviour similar to FORM's `bracket` statement. `HEPCollect[expr, p]` goes through each term in `expr` and finds the factors which contain `p`. It then introduces one level of brackets, combining those terms which have the same set of `p`-dependent factors. The second argument `p` can be a Mathematica pattern, in which case all factors which are not free of this pattern are pulled out.

```

expr = SDot[p, q] a1 + SDot[p, q] a2 +
      Sqr[p] b1 + Sqr[p] b2 +
      SDot[p, q] Sqr[p] c1 + SDot[p, q] Sqr[p] c2;
HEPCollect[expr, p]
⇒ (a1 + a2) SDot[p, q] + (b1 + b2) Sqr[p] +
  (c1 + c2) SDot[p, q] Sqr[p]
HEPCollect[expr, q]
⇒ b1 Sqr[p] + b2 Sqr[p] +
  SDot[p, q] (a1 + a2 + c1 Sqr[p] + c2 Sqr[p])

```

By default `HEPCollect` automatically renames contracted indices to minimise the number of generated terms:

```

HEPCollect[p[mu] k[mu] + p[nu] q[nu], p]
⇒ p[mu] (k[mu] + q[mu])

```

You can disable this feature (e.g. for performance reasons) by setting the option `CombineTensorStructures` to `False`.

HEPCollectListed A common use-case for functions like `HEPCollect` is the task of isolating different colour or Lorentz structures in an expression and extracting the coefficients of these structures. Typically you then want to perform some operations only on the structures or only on the coefficients. The function `HEPCollectListed` is made for this purpose since it returns the structures and coefficients as a pair of lists.

```

expr = a Eta[mu, nu] + b Sqr[p] Eta[mu, nu] + c p[mu] p[nu];
terms = HEPCollectListed[expr, mu|nu]
⇒ {{Eta[mu, nu], p[mu] p[nu]}, {a + b Sqr[p], c}}

```

In fact, the `HEPCollect` function is simply implemented as

```

HEPCollect[expr_, patt_] := Dot @@ HEPCollectListed[expr, patt]

```

Pairs of lists as returned by `HEPCollectListed` can be processed further with the `HEPCoefficient`, `HEPSelectTerms`, and `HEPGroupTerms` functions.

HEPCoefficient can be used to extract the coefficient of a specific term. Continuing the example above, we can extract the coefficient of the `Eta[mu, nu]` term with

```
HEPCoefficient[terms, Eta[mu, nu]]
⇒ a + b Sqr[p]
```

You can use patterns to identify the desired term:

```
HEPCoefficient[terms, _Eta]
⇒ a + b Sqr[p]
```

However, you will receive an error message if the pattern matches more than one term.

HEPSelectTerms Similar to Mathematica's `Select` function you can use `HEPSelectTerms` to filter out those terms for which a given function returns `True`:

```
HEPSelectTerms[terms, Head[#] === Eta &]
⇒ {{Eta[mu, nu]}, {a + b Sqr[p]}}
```

The result is of the same form as the results of `HEPCollectListed`.

HEPGroupTerms If you want to split the terms into two sets according to some criterion you can do that with the `HEPGroupTerms` function. It returns a list of two expressions of the form returned by `HEPCollectListed`.

```
HEPGroupTerms[terms, Head[#] === Eta &]
⇒ {{Eta[mu, nu]}, {a + b Sqr[p]}},
   {{p[mu] p[nu]}, {c}}}
```

In addition to the above-mentioned “generic” tensor manipulation functions `HEP-Math` also provides a few functions which are specific to Lorentz vectors and Dirac matrices.

DiracSubstitute replaces occurrences of `PL`, `PR` and `GSig[mu, nu]` by their definitions in terms of `GI`, `G5` and `Ga[mu]`.

DiracExpand can be used to expand composite factors in `DiracTr` expressions:

```
DiracExpand[DiracTr[(Gs[p] + m GI).PL]]
⇒ (-1/2) m DiracTr[G5] + (1/2) m DiracTr[GI] -
   (1/2) DiracTr[Gs[p].G5] + (1/2) DiracTr[Gs[p]]
```

It is defined as

```
DiracExpand[expr_] :=
  DiracContract[HEPExpand[DiracSubstitute[expr], Gs|Ga|GI|G5]]
```

PullVectors is similar to **PullScalars** but extracts Lorentz vectors from index-free contractions like `Gs[p]` or `SDot[p, q]`. Such an operation is often needed in reduction algorithms for loop integrals.

```
PullVectors[DiracTr[(Gs[p] + m GI).(SDot[p, q] G5)], p]
⇒ DiracTr[(GI m) . G5] p[$3] q[$3] +
   DiracTr[Ga[$4] . G5] p[$4] p[$5] q[$5]
```

The optional second argument can be any Mathematica pattern. Only vectors matching the pattern are pulled out. `PullVectors[expr, p]` expands `expr` in `p`, i.e. it calls `HEPExpand[expr, p]`.

4.2 Index Manipulations and Substitutions

JoinIndexLists can be used to join multiple index lists.

```
JoinIndexLists[USp[p][s][a1] + G5[a1][bt] VSp[q][s][bt]]
⇒ USp[p][s, a1] + G5[a1, bt] VSp[q][s, bt]
```

HEPContract Identity tensors such as `Eta` (for Lorentz indices), `GI` (for Dirac indices) or `ColorDelta` (for colour indices) can be eliminated with the **HEPContract** function. Full contractions of identity tensors with themselves are replaced by the correct dimension.

```
HEPContract[Eta[mu][nu] p[nu]]
⇒ p[mu]
HEPContract[Ga[mu].GI.Ga[nu]]
⇒ Ga[mu] . Ga[nu]
HEPContract[a Eta[mu, mu] + b GI[a1, a1] + c ColorDelta[i, i]]
⇒ 4 b + 3 c + a Dim
```

Note that **HEPContract** invokes **JoinIndexLists**.

LorentzContract, **DiracContract** were already mentioned in Sec. 3. They invoke **HEPContract** and then perform additional replacements specific to the built-in tensors.

ExplicitIndices replaces index-free notations like `SDot[p, q]` with explicit index contractions. It only works on expressions without suppressed indices.

ExplicitSlash writes the contraction `Gs[p]` with explicit indices. This function also works on expressions with suppressed indices:

```
ExplicitSlash[Gs[p]]
⇒ Ga[$3] p[$3]
```

HEPMultiply When dealing with tensor expressions name clashes between contracted indices are a common problem. Consider the expressions $p[\mu] k[\mu]$ and $q[\mu] k[\mu]$. By themselves they are absolutely fine, but if you want to multiply them you have to re-name the contracted index μ in one of them. For complicated expressions the **Indices** function mentioned in Sec. 3 can help you find all contracted indices. Alternatively, you can let HEPMath do the renaming automatically by using the **HEPMultiply** function. It replaces contracted indices in all its arguments by unique symbols and then multiplies them.

```
HEPMultiply[p[mu] k[mu], q[mu] k[mu]]
⇒ k[$3] k[$4] p[$3] q[$4]
```

A common application is squaring a Feynman amplitude.

```
squaredme = HEPMultiply[amp, CC[amp]]
```

HEPReplaceAll Another case where name clashes between contracted indices can occur are substitutions. Take the expression $a p[\mu] k[\mu]$. Replacing a with $q[\mu] k[\mu]$ using Mathematica's **ReplaceAll** function (also known as the `/.` operator) will give you an invalid expression. HEPMath can take care of this if you use the **HEPReplaceAll** function instead.

```
HEPReplaceAll[p[mu] k[mu] a, a -> q[mu] k[mu]]
⇒ k[mu] k[$3] p[mu] q[$3]
```

ReplaceIndices If you simply want to replace all contracted indices in an expression with unique symbols you can use the **ReplaceIndices** function.

```
ReplaceIndices[p[mu] k[mu]]
⇒ p[$3] k[$3]
```

HEPSetZero Setting a tensor variable to zero, either globally or by replacement rules, can lead ill-formed expressions. For example, the replacement $p \rightarrow 0$ transforms the expression $p[\mu] k[\mu]$ into the somewhat undesirable result $0[\mu] k[\mu]$. Unfortunately there is no way to tell Mathematica to automatically simplify such expressions to zero. **HEPSetZero**[*expr*, *p*] sets the tensor p to zero inside the expression *expr* without generating ill-formed expressions and returns the result:

```
HEPSetZero[Sqr[k] + p[mu] k[mu], p]
⇒ Sqr[k]
```

The second argument can also be a pattern, in which case all tensors matching that pattern are set to zero.

4.3 Tensor Derivatives

Derivatives with respect to tensor variables can be computed with the `HEPD` function. It mimicks the behaviour of Mathematica's `D` function.

```
HEPD[expr, t[i1,...,in]]
```

returns the derivative of expression `expr` with respect to the tensor variable `t` with indices `i1,...,in`. This derivative would usually be denoted

$$\frac{\partial \text{expr}}{\partial t_{i_1 \dots i_n}} \quad .$$

The index list must be complete, i.e. the expression `t[i1,...,in]` must have tensor signature `{}`. The indices `i1,...,in` of the derivative must be different from any contracted indices in `expr`. They can, however, be the same as an external index of `expr`, in which case the derivative index is contracted with the external index of `expr`. They can also be contracted with each other. Higher derivatives can be computed by simply adding arguments to `HEPD`

```
HEPD[expr, t[i1,...,in], r[j1,...,jm], ...]
```

The `HEPD` function can handle sums, products, powers, and index-free contractions (via `SDot` and `Sqr`) of arbitrary tensor expressions. More complicated constructs such as `Gs` or `DiracTr` are currently not supported. (But you can always use `ExplicitIndices` to obtain an expression that `HEPD` can handle.) When `HEPD` encounters an expression it cannot differentiate it returns a `HEPDerivative` expression:

```
HEPD[Gs[p][al, bt], p[mu]]
⇒ HEPDerivative[Gs[p][al, bt], p[mu]]
```

Thus, tensor derivatives of user-defined functions can be set by specifying rules for `HEPDerivative`.

4.4 Simplification of Colour Structures

To simplify colour structures and compute colour factors you can use the `ColorReduce` function. It replaces occurrences of `ColorF` with appropriate traces over the generators T^a and then applies the identity

$$T_{ij}^a T_{kl}^b = \frac{1}{2} \delta_{il} \delta_{kj} - \frac{1}{6} \delta_{ij} \delta_{kl} \quad .$$

Note that it only operates on expressions with no suppressed colour or gluon indices.

```

ColorReduce[ColorF[a, b, c]]
⇒ (-2 I) ColorTr[ColorT[a] . ColorT[b] . ColorT[c]] +
   (2 I) ColorTr[ColorT[a] . ColorT[c] . ColorT[b]]
ColorReduce[(ColorT[a].ColorT[a])[i, j]]
⇒ (4/3) ColorDelta[i, j]
ColorReduce[ColorT[a].ColorT[a]]
⇒ ColorReduce::nosup: ColorReduce only works on expressions
   without suppressed ColorFundamental and ColorAdjoint indices.
⇒ ColorT[a] . ColorT[a]
ColorReduce[ColorF[a, b, c] ColorF[a, b, c]]
⇒ 24

```

4.5 Polarisation Sums

The summation of a squared amplitude over the polarisations of external fermions or vector bosons can be done analytically with the `FermionPolarizationSums` and `VectorPolarizationSums` functions, respectively. The `FermionPolarizationSums` function uses the formulae

$$\sum_s u(p, s)\bar{u}(p, s) = \not{p} + m \quad , \quad \sum_s v(p, s)\bar{v}(p, s) = \not{p} - m \quad .$$

The masses associated with the external momenta are specified in the second argument by a rule or a list of rules. Here are some examples:

```

FermionPolarizationSums[Bar[USp[p]][s] . Ga[mu] . USp[p][s], p -> m]
⇒ DiracTr[(GI*m + Gs[p]) . Ga[mu]]
FermionPolarizationSums[Bar[USp[p]][s, al] USp[p][s, bt], p -> m]
⇒ (GI*m + Gs[p])[al, bt]
FermionPolarizationSums[
  Bar[USp[p]][r].VSp[k][s] Bar[VSp[k]][s].USp[p][r],
  {p -> m, k -> 0}]
⇒ DiracTr[(GI*m + Gs[p]) . Gs[k]]

```

Note that you can use patterns in the rules specifying the masses.

```

DeclareLorentzVectors[_q];
FermionPolarizationSums[
  Bar[USp[q[1]]][s] . Ga[mu] . USp[q[1]][s], q[n_] -> m[n]]
⇒ DiracTr[(Gs[q[1]] + GI*m[1]) . Ga[mu]]

```

If the mass associated with a certain external momentum is not specified the corresponding spinors are left alone.

```

FermionPolarizationSums [
  Bar[USp[p]][r].VSp[k][s] Bar[VSp[k]][s].USp[p][r], k -> 0]
=> Bar[USp[p]][r] . Gs[k] . USp[p][r]

```

The function `VectorPolarizationSums` works in a similar way. It applies the formulae

$$\sum_s \varepsilon_\mu^*(k, s) \varepsilon_\nu(k, s) = -g_{\mu\nu} + \frac{k_\mu k_\nu}{m^2} \quad \text{for massive particles,}$$

$$\sum_s \varepsilon_\mu^*(k, s) \varepsilon_\nu(k, s) = -g_{\mu\nu} - \frac{n^2 k_\mu k_\nu}{(n \cdot k)^2} + \frac{n_\mu k_\nu + n_\nu k_\mu}{n \cdot k} \quad \text{for massless particles.}$$

The polarisation sum for massless vector bosons depends on a *gauge vector* n which must satisfy $n \cdot \varepsilon(k, s) = 0$ and $n \cdot k \neq 0$. The masses of massive vector bosons and the gauge vectors of massless gauge bosons can be passed to `VectorPolarizationSums` as a list of rules, similar to the Fermion case:

```

VectorPolarizationSums[CC[MPol[k]][s, mu] MPol[k][s, nu], k -> m]
=> -Eta[mu, nu] + (k[mu]*k[nu])/m^2
DeclareLorentzVectors[n];
VectorPolarizationSums[CC[Pol[k]][s, mu] Pol[k][s, nu], k -> n]
=> -Eta[mu, nu] + (k[nu]*n[mu] + k[mu]*n[nu])/SDot[k, n] -
  (k[mu]*k[nu]*Sqr[n])/SDot[k, n]^2

```

As in the case of `FermionPolarizationSums` you can use patterns in the second argument and polarisation vectors are left alone if no mass or gauge vector are specified for the external momentum.

You can apply an arbitrary function to the denominators of massless polarisation sums with the `PostProcess` option. This is useful for performing simplifications specific to your choice of the gauge vector. For example, if you have chosen n so that $n \cdot k = 1$ you could implement this with

```

VectorPolarizationSums[CC[Pol[k]][s, mu] Pol[k][s, nu], k -> n,
  PostProcess -> ((# /. SDot[k, n] -> 1)&)]
=> -Eta[mu, nu] + k[nu]*n[mu] + k[mu]*n[nu] - k[mu]*k[nu]*Sqr[n]

```

4.6 One-Loop Integrals

A common technique in the computation of one-loop amplitudes is to express the loop integrals in terms of *Passarino-Veltman* (tensor) functions. In HEPMath this step is automated with the `PaVeIntegrate` function. Its syntax is

`PaVeIntegrate[expr, l, muR]`

where `expr` is the expression to integrate, `l` is the loop momentum to integrate over and `muR` is the renormalisation scale. This computes the integral

$$(2\pi\mu R)^{4-D} \int d^D l \text{ expr}$$

where $D \equiv \text{Dim}$ is the dimension of Minkowski space. Propagator denominators in `expr` must be represented by `Den` expressions (see Tab. 1). `PaVeIntegrate` invokes `PullVectors[expr, l]` and then replaces products of loop momenta and propagator denominators with the correct linear combinations of Passarino-Veltman one-loop tensor integrals. These integrals are represented as follows

`PaVe[n, muR, irpow][indices][invariants]`

where `n` is the number of external legs (or propagator denominators) of the integral, `muR` the renormalisation scale, `indices` the sequence of indices identifying the tensor integral and `invariants` the sequence of kinematic invariants specifying the external momenta and internal masses. Currently only integrals with up to four propagator denominators are supported. The argument `irpow` denotes the negative power of the infrared regulator ε_{IR} in dimensional regularisation. Its possible values are 0, 1 and 2.

For example, the two-point rank two tensor integral can be computed as follows:

```
DeclareLorentzVectors[l, p];
DeclareReal[m0, m1];
PaVeIntegrate[l[mu] l[nu] Den[l, m0] Den[l+p, m1], l, muR]
⇒ I*Pi^2*(Eta[mu, nu]*(Div*((m0^2 + m1^2)/2 - Sqr[p]/12) +
    PaVe[2, muR, 0][0, 0][Sqr[p], m0^2, m1^2]) +
    p[mu]*p[nu]*(Div/3 + PaVe[2, muR, 0][1, 1][Sqr[p], m0^2, m1^2]))
```

The symbol `Div` represents the UV divergence

$$\text{Div} = \frac{2}{4 - \text{Dim}} - \gamma_E + \ln(4\pi)$$

where γ_E is the Euler constant. The tensor integral

`PaVe[2, muR, 0][1, 1][Sqr[p], m0^2, m1^2]`

would usually be denoted as $B_{11}(p^2, m_0^2, m_1^2)$ with the dependence on `muR` implicit. For the indices and order of arguments of the tensor integrals `HEPMath` uses the same conventions as `LoopTools`, and I refer the reader to the `LoopTools` manual for details.

By default `PaVeIntegrate` only returns the IR finite part of the integral. If you set the option `IRDivergentParts` to `True` the result will be given as a power series in `IRDiv`, which represents the IR divergence $1/\varepsilon_{\text{IR}}$.

```

PaVeIntegrate[Den[1, 0] Den[1+p, 0], 1, muR, IRDivergentParts->True]
=> I*Pi^2*(Div + PaVe[2, muR, 0][0][Sqr[p], 0, 0] +
  IRDiv*PaVe[2, muR, 1][0][Sqr[p], 0, 0] +
  IRDiv^2*PaVe[2, muR, 2][0][Sqr[p], 0, 0])

```

4.7 Feynman Parametrisation of Multi-Loop Integrals

The standard method for the (analytic) computation of one-loop or multi-loop integrals is to trade the integral over the loop momenta for an integral over *Feynman parameters*. In HEPMath you can do this transformation automatically with the `FeynmanIntegrate` function. It only works for *scalar integrals*, i.e. for integrals without any loop momenta appearing in the numerator. This means that before using `FeynmanIntegrate` you need to cancel any appearances of loop momenta in the numerator against the propagator denominators. The `DenCancel` function can help you with that. It will cancel any *squared* momenta in the numerator against propagator denominators with the same momentum. However, it can not (yet) eliminate scalar products between loop and external momenta or between different loop momenta from the numerator. The syntax is

```
DenCancel[expr, patt]
```

This will cancel squares of momenta in `expr` which are not free of the pattern `patt` against propagator denominators (`Den` expressions) with the same momentum. If you omit the second argument it will try to cancel all squared momenta. Note that the sign ambiguity of the arguments of `Sqr` and `Den` is automatically taken care of:

```
DenCancel[Sqr[-1] Den[1, m], 1]
=> 1 + m^2 Den[1, m]
```

The syntax of `FeynmanIntegrate` is

```
FeynmanIntegrate[expr, {l1, ..., lL}, x, muR, del]
```

where `expr` is the integrand expression, `{l1, ..., lL}` is the list of loop momenta, `muR` is the renormalisation scale, and `del` is the symbol used for the infinitesimal imaginary part of the propagators. The argument `x` is the head used to generate the Feynman parameters `x[1]`, `x[2]`, etc. You can omit the `del` argument, in which case it is set to zero. `FeynmanIntegrate` computes the integral

$$(2\pi\mu R)^{L(4-\text{Dim})} \int d^{\text{Dim}} l_1 \dots \int d^{\text{Dim}} l_L \text{expr} \quad ,$$

where L is the number of loop momenta. The resulting Feynman integrals are of the form

$$\mathcal{G}(D) \int_0^1 dx_1 \cdots \int_0^1 dx_n \delta(1 - x_1 - \dots - x_n) \mathcal{M}(\mathbf{x}) \mathcal{F}(\mathbf{x})^{\alpha(D)} \mathcal{U}(\mathbf{x})^{\beta(D)} \quad , \quad (1)$$

where $\mathcal{G}(D)$ is a pre-factor involving Gamma functions and the space-time dimension $D(\equiv \text{Dim})$, $\mathcal{M}(\mathbf{x})$ is a monomial in the Feynman parameters x_i originating from denominators raised to some power, $\mathcal{F}(\mathbf{x})$ and $\mathcal{U}(\mathbf{x})$ are polynomials in the x_i and $\alpha(D)$ and $\beta(D)$ are powers depending on D . In HEPMath these integrals are represented by `FeynmanIntegral` expressions, which have the following form:

$$\text{FeynmanIntegral}[\{\mathcal{G}(D), \mathcal{M}(\mathbf{x}), \mathcal{F}(\mathbf{x})^{\alpha(D)}, \mathcal{U}(\mathbf{x})^{\beta(D)}\}, \{x_1, \dots, x_n\}]$$

The first thing you usually want to do with these integrals is integrate out the delta function by eliminating one Feynman parameter. (Since you may want to have a say in which Feynman parameter gets eliminated HEPMath doesn't do this automatically.) This can be done with the `EliminateFeynmanParameter` function:

```
EliminateFeynmanParameter[f, x[i]]
```

integrates the delta function by eliminating the i -th Feynman parameter. If you drop the second argument the parameter is chosen automatically. The argument f must be a `FeynmanIntegral` expression, so you'll typically use a replacement rule like

```
expr = expr /. f_FeynmanIntegral -> EliminateFeynmanParameter[f]
```

The expressions returned by `EliminateFeynmanParameter` still contain integrals over the remaining Feynman parameters. These integrals are represented by `HEPIntegral` expressions. `HEPIntegral` has the same syntax as Mathematica's `Integrate` function, but it does not attempt to compute the integrals. It is just a container for intermediate expressions which you will need to manipulate further to compute the integrals. However, you can expand `HEPIntegral` expressions with `HEPExpand`.

5 User-Defined Tensors and Linear Functions

Pre-defined tensors such as `Ga` or `Eps` are actually just examples of HEPMath's capability of integrating arbitrary tensors and index types in Mathematica. To see how this works let us try to add the notion of Weyl spinors to HEPMath. Note that the code in this section is just an example. Weyl spinors are (currently) *not* included in HEPMath.

First we have to add an index type which represents Weyl spinors. (For simplicity we will not distinguish between dotted, undotted, raised and lowered indices). New index types can be declared with the `DeclareIndexType` function:

```
DeclareIndexType[Weyl, 2, WI]
```

This introduces a new index type called `Weyl` with index dimension 2. It also tells `HEPMath` that the identity matrix in the space of Weyl spinors is represented by `WI`. Note that `HEPContract` immediately knows what to do with `WI`:

```
HEPContract[WI[aa, bb] WI[bb, cc]]
⇒ WI[aa, cc]
HEPContract[WI[aa, aa]]
⇒ 2
```

Next we introduce the σ^μ and $\bar{\sigma}^\mu$ matrices and the anti-symmetric tensor in Weyl-space:

```
DeclareHEPTensor[{WSig, WSigBar}, {Lorentz, Weyl, Weyl}]
DeclareHEPTensor[WEps, {Weyl, Weyl}]
```

Functions like `HEPTensorSignature`, `Indices`, `ExplicitIndices` or `HEPExpand` now accept the new objects as first-class citizens.

```
DeclareLorentzVectors[p];
HEPTensorSignature[p.WSig]
⇒ {Weyl, Weyl}
Indices[WSig[mu, aa, bb] WSigBar[mu, bb, cc]]
⇒ {{aa -> Weyl, cc -> Weyl}, {bb -> Weyl, mu -> Lorentz}}
ExplicitIndices[(p.WSig)[aa,bb]]
⇒ p[$3]*WSig[$3, aa, bb]
HEPExpand[(p.(a WSig + b WSigBar))[aa,bb]]
⇒ a*(p . WSig)[aa, bb] + b*(p . WSigBar)[aa, bb]
```

We may also want to tell `HEPMath` that `WEps` is real.

```
DeclareReal[WEps];
CC[WEps[aa, bb]]
⇒ WEps[aa, bb]
```

Note that there is currently no general way to declare a tensor as anti-symmetric. To exploit this property you will have to write your own simplification function.

Traces over Dirac and colour indices are represented by `DiracTr` and `ColorTr`, respectively, and we may want to have a similar notation for Weyl indices. We therefore introduce a new symbol `WeylTr` which represents a trace over Weyl indices. There are two properties of `WeylTr` which `HEPMath` needs to know about: the trace is *linear* and *real holomorphic*, meaning that the complex conjugate of a trace is the same as the trace of the complex conjugate of its argument. These two properties can be declared separately with

```

DeclareLinear[WeylTr];
DeclareRealHolomorphic[WeylTr];
HEPEExpand[WeylTr[(a WSig + b WSigBar)[mu]]]
⇒ a*WeylTr[WSig[mu]] + b*WeylTr[WSigBar[mu]]
CC[WeylTr[WSig[mu] + p[mu] WEps]]
⇒ WeylTr[WEps*p[mu] + CC[WSig][mu]]

```

Note that declaring a function as linear means that it is linear in *all its arguments*. `DeclareLinear[WeylTr]` not only instructs `HEPEExpand` how to operate on `WeylTr` expressions, it also tells `HEPMath` that `WeylTr` “exports” any free or contracted indices appearing in its argument:

```

Indices[WeylTr[p[mu] WSig[mu].WSigBar[nu]]]
⇒ {{nu -> Lorentz}, {mu -> Lorentz}}

```

If a function is not declared as linear `HEPMath` will simply ignore any indices appearing in its arguments:

```

Indices[somefunc[p[mu] WSig[mu].WSigBar[nu]]]
⇒ {}, {}

```

6 Interfaces

In cases where many different tools are used in the same project the development of interfaces between these tools can take a substantial amount of time. With its consistent implementation of High Energy Physics notations in Mathematica, `HEPMath` is the ideal environment for combining different tools related to High Energy Physics. `HEPMath` currently provides interfaces to `FeynArts`, `LoopTools` and `LHAPDF`, but I hope that this list will get longer in the future. This section describes the interfaces currently present in `HEPMath`.

6.1 FeynArts

The `FeynArts` interface of `HEPMath` (obviously) requires a working `FeynArts` installation. `HEPMath` was tested with `FeynArts` version 3.8. To use `FeynArts` from within `HEPMath` you must *not* load the `FeynArts` context directly. Instead you call

```
Needs["HEPMath`FeynArts`"]
```

This will give you access to all FeynArts functions and symbols, but it moves some of the FeynArts symbols to a different context in order to avoid name clashes with HEPMath.

After loading the FeynArts interface you can then use FeynArts in the same way as described in the FeynArts manual, except for one small difference: if you need access to symbols defined in your FeynArts model file you *must* initialise the model before these symbols are mentioned. The reason is that HEPMath tweaks FeynArts² so that it does not load models into the global context but in the context `HEPMath`FeynArts`Model` instead. This way the global context does not get polluted and name clashes between symbols in the model file and HEPMath symbols can be avoided. To make sure that all symbols from the model file end up in `HEPMath`FeynArts`Model` and not `Global`` you have to initialise the model before using any of its symbols. Here is an example of what can go wrong:

```
Needs["HEPMath`"];
Needs["HEPMath`FeynArts`"];
(* InitializeModel["SM"]; *)
SetOptions[InsertFields, Model -> "SM",
           InsertionLevel -> {Particles},
           Restrictions -> NoLightFHCoupling];
process = {S[1]} -> {F[4, {3}], -F[4, {3}]}];
topos = CreateTopologies[0, 1 -> 2];
ins = InsertFields[topos, process];
amp = CreateFeynAmp[ins];
=> InitializeModel::badrestr:
      Warning: Global`NoLightFHCoupling is not a valid model
      restriction.
```

Removing the comments around `InitializeModel["SM"];` gets rid of the warning and everything will run smoothly.

To manipulate the amplitude `amp` from the example above in HEPMath you first have to convert it to HEPMath notation. For the model files included in the FeynArts distribution this can be done with the `ConvertFeynAmp` function. Continuing the example above you can call

```
{cfg, amp} = ConvertFeynAmp[amp];
cfg
=> {Mass[Incoming, 1] -> MH, Mass[Outgoing, 1] -> MT,
     Mass[Outgoing, 2] -> MT, Indices[Incoming, 1] -> {},
     Indices[Outgoing, 1] -> {Index[Colour, 2]}}
```

²Not permanently. The FeynArts source files are not modified and loading FeynArts without HEPMath will give you the usual behaviour.

```

Indices[Outgoing, 2] -> {Index[Colour, 3]}
amp
⇒ I*ColorDelta[Index[Colour, 2], Index[Colour, 3]]*
  Bar[USp[FourMomentum[Outgoing, 1]]][Polarization[Outgoing, 1]] .
  ((-I/2*EL*MT*PL)/(MW*SW) - (I/2*EL*MT*PR)/(MW*SW)) .
  VSp[FourMomentum[Outgoing, 2]][Polarization[Outgoing, 2]]

```

`cfg` is a list which holds information about the masses and indices of the external particles and `amp` is an expression which represents the Feynman amplitude. `Polarization[Outgoing, 1]` denotes the spin polarisation index of the first outgoing particle (i.e. the top quark). Note that the `FourMomentum` expressions are recognised as Lorentz vectors by `HEPMath`.

```

HEPTensorSignature[FourMomentum[Outgoing, 1]]
⇒ {Lorentz}

```

Thus the expression `amp` can be directly manipulated with the `HEPMath` functions described in Sec. 4. To make intermediate results easier to read I usually introduce some shorthands like

```

DeclareLorentzVectors[_k, _p];
amp = amp /. {FourMomentum[Incoming, n_] -> k[n],
             FourMomentum[Outgoing, n_] -> p[n],
             FourMomentum[Internal, n_] -> l[n],
             Polarization[Incoming, n_] -> sg[n],
             Polarization[Outgoing, n_] -> lm[n],
             Index[Gluon, n_] -> a[n],
             Index[Colour, n_] -> i[n],
             Index[Lorentz, n_] -> mu[n]}
⇒ I*ColorDelta[i[2], i[3]]*Bar[USp[p[1]]][lm[1]] .
  ((-I/2*EL*MT*PL)/(MW*SW) - (I/2*EL*MT*PR)/(MW*SW)) .
  VSp[p[2]][lm[2]]

```

but this is purely a matter of taste.

The function `ConvertFeynAmp` should work for amplitudes generated with one of the model files shipping with `FeynArts`. However, since `FeynArts` does not really define the set of kinematic objects which are allowed to appear in model files (and thus in expressions returned by `CreateFeynAmp`) it is not possible to write a conversion function that is guaranteed to work with any model file. If you have a model file that is not supported by `HEPMath` and you think it should be free to let me know.

6.2 LoopTools

The LoopTools package [2] allows you to evaluate one-loop tensor integrals numerically. If you enabled the LoopTools interface during the installation you can load it in your Mathematica session with

```
Needs["HEPMath`LoopTools`"]
```

The PaVe expressions introduced in Sec. 4.6 will now evaluate to numbers when you call them with numeric arguments.

```
PaVe[2, 1.0, 0][0][10.0, 1.0, 2.0]
⇒ 0.622592 + 2.0116 I
```

If you want to generate numerical code which calls functions from LoopTools you also need the LoopTools interface installed.

6.3 LHAPDF

The LHAPDF library is a common C++ interface for a large number of parton distribution functions (PDFs). HEPMath provides a simple interface to LHAPDF which lets you load PDF sets and evaluate PDFs numerically in Mathematica. As in the case of LoopTools you need to enable the LHAPDF interface during installation to use it in Mathematica, and also to generate code which calls LHAPDF functions.

You can load the interface with

```
Needs["HEPMath`LHAPDF`"]
```

To load a specific PDF set you call, for example,

```
pdfid = LHAPDFOpen["cteq611", 0]
```

The first argument identifies the collection of PDFs you want to use and the second argument is the number of the desired PDF set within that collection. The function loads the PDF set into Mathematica and returns the “LHAPDF-ID”, a number which uniquely identifies that set. If you know that number you can also load the set with

```
LHAPDFOpenID[pdfid]
```

To get the number of PDF sets in a given collection (and thus the allowed range of values for the second argument of LHAPDFOpen) you can call

```
LHAPDFMembers["cteq611"]  
⇒ 1
```

The `cteq611` collection only contains one set.

Once you have loaded the PDF set you want and stored its ID in `pdfid` you can evaluate the PDF with

```
LHAPDF[pdfid, pid, x, Q]
```

Here `pid` is the PDG code of the parton, `x` is the momentum fraction and `Q` the factorisation scale. To get a list of PDG codes of the partons included in the PDF set call

```
LHAPDFFlavors[pdfid]
```

To compute the value of α_s associated with the PDF set at the scale `Q` use

```
LHAPDFAlphaS[pdfid, Q]
```

When you are done with the PDF set remember to close it with

```
LHAPDFClose[pdfid]
```

7 Code Generation

More often than not the result of an analytical calculation can not be simplified to a digestible form, and typically some extensive numerical procedure needs to be applied to the result. For cross sections this procedure is usually the phase space integration. In other cases one might want to include the result in a fit where the computed quantity contributes to a χ^2 function which is minimised numerically.

Of course Mathematica is perfectly capable of doing numerical calculations. The efficiency of these calculations can be significantly increased with Mathematica's `Compile` function, which compiles Mathematica expressions into byte-code. With the recently added support for native compilation targets the performance is actually comparable to hand-written C code. However, due to the limited availability of Mathematica licenses at most institutes one needs a different solution when the numerical computation is to be parallelised and run on a computer cluster.

The conventional approach is to use Mathematica's `CForm` or `FortranForm` functions to convert expressions into C or Fortran code snippets which are then pieced together

and compiled outside Mathematica. The disadvantages of this approach are that it is not very general and that any code optimisation which is beyond the compiler's capabilities must be implemented by the user (or by the package providing the code generation functionality, as in the case of FormCalc). Since Mathematica version 8 there is actually a viable alternative. If the function only uses a certain subset of *compilable* Mathematica functions the byte-code generated by Mathematica's `Compile` function can now be exported to C code. `Compile` does a rather decent job at optimising large expressions, and since the C code is generated from byte-code rather than Mathematica expressions it benefits from the optimisation done by `Compile`. This method of generating code is not only efficient but also very flexible. You can first use Mathematica to write, test and compile a function which does exactly what you want, and then put the compiled function into a shared library which you can link with the rest of your code. The code generation system implemented in HEPMath is based on this strategy.

Mathematica's code generation from byte-code currently has a major shortcoming: it is not possible for the generated code to call functions from other libraries. HEPMath provides a workaround for this by post-processing the generated code. As a result the code generated by HEPMath can call any external function which is interfaced to Mathematica via the `LibraryLink` interface.

HEPMath provides a function `HEPCompile` which mirrors the behaviour of Mathematica's `Compile` function but guarantees the correct treatment of symbols introduced by HEPMath. The use of `HEPCompile` is best explained by examples:

```
Needs["HEPMath`Compile`"];
expr = x^2 + y^2;
cf1 = HEPCompile[{{x, _Real}, {y, _Real}},
  HEPEvaluate[expr] + 1.0];
cf1[1.0, 1.0]
⇒ 3.
```

The first argument to `HEPCompile` specifies the names and types of arguments to the function. Its syntax is identical to the one used in `Compile`. The second argument is the function body. Like `Compile`, `HEPCompile` does not evaluate the function body. (It has the attribute `HoldAll`.) However, since you usually want to use `HEPCompile` to compile large expressions and you probably don't want to type them in by hand you can selectively evaluate parts of the function body with `HEPEvaluate`. Wrapping `HEPEvaluate` around an expression inside a `HEPCompile` function body instructs `HEPCompile` to substitute the evaluation of that expression before compilation. The body of the above function is therefore $x^2 + y^2 + 1.0$ and thus evaluates to 3.0 when called with $x=y=1.0$.

`HEPCompile` can also handle function bodies that depend on Lorentz vectors and contain one-loop integrals. In the function body of `HEPCompile` Lorentz vectors are represented by lists of four numbers (the first element being the time component).

However, before injecting symbolic expressions which contain Lorentz vectors or PaVe expressions in a function body some additional manipulations are necessary. You can let `HEPCCompile` do these manipulations automatically by wrapping `HEPPPrepare` around the relevant parts of the function body:

```
DeclareLorentzVectors[p];
b0 = PaVe[2, 1.0, 0][0][Sqr[p], m^2, m^2];
cf2 = HEPCCompile[{{p, _Real, 1}, {m, _Real}},
  HEPPPrepare[b0]];
cf2[{5.0, 0.0, 0.0, 0.0}, 2.0]
⇒ -0.218071 + 1.88496 I
```

In addition to `HEPEvaluate`, `HEPPPrepare` and the compilable functions supported by `Compile` (call Mathematica's undocumented function `Compile[CompilerFunctions[]` for a list) you can use the following expressions inside a `HEPCCompile` body:

- `Pol[p][s]` and `MPol[p][s]` return the polarisation vectors of a massless and massive vector boson with momentum p and helicity s . The argument p and the return values are lists of four numbers and s must be an integer which can be -1 or 1 for `Pol` and -1, 0 or 1 for `MPol`.
- `LHAPDFOpenID`, `LHAPDF`, `LHAPDFAlphaS` and `LHAPDFClose` with their correct argument lists.
- `KineticLambda[x,y,z]` represents the kinetic function $\lambda(x, y, z) = x^2 + y^2 + z^2 - 2(xy + zy + zx)$.
- `TwoBodyDecay[q, m1, m2, theta, phi]` can be used for constructing phase space parametrisations. It returns a pair of Lorentz vectors (i.e. lists of four numbers) which are the momenta of the decay products of a particle with four momentum q decaying into two particles with masses $m1$ and $m2$. The arguments `theta` and `phi` are polar and azimuthal angles in the rest frame of the decaying particle.

Once you have compiled all the functions you need you can export them with the `HEPCCodeGenerate` function. It mirrors the behaviour of Mathematica's `CCodeGenerate` function but should only be used on compiled functions generated with `HEPCCompile`. To export the functions `cf1` and `cf2` from the examples above to a library called `myfuncs` execute

```
modules = HEPCCodeGenerate[{cf1, cf2}, {"f1", "f2"}, "myfuncs",
  TargetDirectory->"build"]
⇒ {"build/myfuncs.c", "build/myfuncs_f1.c", "build/myfuncs_f2.c"}
```

This puts a number of files in the directory `build` and returns a list with the names of the C source files it generated. Make sure the directory `build` exists before you run this. You will also find a file called `myfuncs.py` in that directory. This is the python wrapper which will allow you to call your functions from Python. But first you have to compile the sources into a shared library. You can do this directly from Mathematica by using the `CCompilerDriver` package:

```
Needs["CCompilerDriver"];
CreateLibrary[modules, "_myfuncs",
  "TargetDirectory" -> "build",
  "Debug"->True, "Libraries"->{"WolframRTL", "dl"},
  "ShellCommandFunction"->Print]
```

The advantage of this method is that all the flags needed to compile against the Wolfram run-time library will be set automatically by Mathematica. You can also do this manually, e.g. in a makefile. In that case simply copy the flags from the output generated by the `CreateLibrary` call above. Note that the first argument to `CreateLibrary` is the list of modules returned by `HEPCodeGenerate`. The second argument is the name of your library (without the extension). If you want to use the Python interface this should always be the name specified in `HEPCodeGenerate` preceded by an underscore. Also note that you have to specify the Wolfram run-time library and the `dl` library in the `"Libraries"` option. The latter is needed for external calls to other libraries, e.g. to `LoopTools`.

If everything worked you should now have a file `_myfuncs.so` in your `build` directory. To use this library you first have to make sure that the Wolfram run-time library is found by your linker. This can be accomplished by adding the location of `libWolframRTL.so` to your `LD_LIBRARY_PATH` variable. On my system this location is

```
/usr/local/Wolfram/Mathematica/9.0/SystemFiles/Libraries/Linux-x86-64
```

but it might be somewhere else on yours. When this is done you can go into the `build` directory, open a Python session and run

```
>>> import myfuncs
>>> myfuncs.f1(1.0, 1.0)
3.0
>>> myfuncs.f2([5.0, 0.0, 0.0, 0.0], 2.0)
(-0.21807097779182483+1.8849555921538759j)
```

The Lorentz vectors can be passed as lists or as `numpy` arrays. If your function would return a list or a matrix in Mathematica this will also become a `numpy` array in Python.

To have the `myfuncs` module available in all your Python sessions just move the files `myfuncs.py` and `_myfuncs.so` to some place where Python will find them (e.g. a directory listed in your `PYTHONPATH` variable). Just make sure that the wrapper module (`myfuncs.py`) and the library (`_myfuncs.so`) are always kept in the same directory.

8 Conclusions

This concludes our tour of the Mathematica package *HEPMath*. Its goal is not to implement a particular computational algorithm or set of algorithms but to make the implementation of symbolic computations related to High Energy Physics easier. It accomplishes this with two features which set it apart from similar public codes: a consistent and extensible integration of tensors and index contractions in the Mathematica language and a flexible mechanism for generating numerical code which allows the direct transition between two high-level programming languages: Mathematica and Python. In addition it provides seamless interfaces to three popular tools: FeynArts, LoopTools and LHAPDF. It still lacks some of the more advanced features present in other packages like the Passarino-Veltman reduction to scalar integrals or the automatic computation of renormalisation constants. I hope to remedy this situation in future releases. *HEPMath* was recently used for a computation of Higgs+jet cross sections in the presence of effective dimension six operators [36].

Acknowledgements

I would like to thank Ulrich Nierste, Stefan Schacht and Wolfgang Noll for helpful comments and beta-testing.

References

- [1] T. Stelzer and W. Long, *Comput.Phys.Commun.* **81** (1994) 357–371, [arXiv:hep-ph/9401258](#) [hep-ph].
- [2] T. Hahn and M. Perez-Victoria, *Comput.Phys.Commun.* **118** (1999) 153–165, [arXiv:hep-ph/9807565](#) [hep-ph].
- [3] F. Yuasa, J. Fujimoto, T. Ishikawa, M. Jimbo, T. Kaneko, *et al.*, *Prog.Theor.Phys.Suppl.* **138** (2000) 18–23, [arXiv:hep-ph/0007053](#) [hep-ph].
- [4] A. Kanaki and C. G. Papadopoulos, *Comput.Phys.Commun.* **132** (2000) 306–315, [arXiv:hep-ph/0002082](#) [hep-ph].
- [5] M. Moretti, T. Ohl, and J. Reuter, [arXiv:hep-ph/0102195](#) [hep-ph].

- [6] F. Krauss, R. Kuhn, and G. Soff, *JHEP* **0202** (2002) 044, [arXiv:hep-ph/0109036 \[hep-ph\]](#).
- [7] **CompHEP** Collaboration, E. Boos *et al.*, *Nucl.Instrum.Meth.* **A534** (2004) 250–259, [arXiv:hep-ph/0403113 \[hep-ph\]](#).
- [8] A. Cafarella, C. G. Papadopoulos, and M. Worek, *Comput.Phys.Commun.* **180** (2009) 1941–1955, [arXiv:0710.2427 \[hep-ph\]](#).
- [9] W. Kilian, T. Ohl, and J. Reuter, *Eur.Phys.J.* **C71** (2011) 1742, [arXiv:0708.4233 \[hep-ph\]](#).
- [10] J. Alwall, P. Demin, S. de Visscher, R. Frederix, M. Herquet, *et al.*, *JHEP* **0709** (2007) 028, [arXiv:0706.2334 \[hep-ph\]](#).
- [11] T. Gleisberg, S. Hoeche, F. Krauss, M. Schonherr, S. Schumann, *et al.*, *JHEP* **0902** (2009) 007, [arXiv:0811.4622 \[hep-ph\]](#).
- [12] A. Belyaev, N. D. Christensen, and A. Pukhov, *Comput.Phys.Commun.* **184** (2013) 1729–1769, [arXiv:1207.6082 \[hep-ph\]](#).
- [13] A. Semenov, *Nucl.Instrum.Meth.* **A389** (1997) 293–294.
- [14] A. Semenov, *Comput.Phys.Commun.* **180** (2009) 431–454, [arXiv:0805.0555 \[hep-ph\]](#).
- [15] N. D. Christensen and C. Duhr, *Comput.Phys.Commun.* **180** (2009) 1614–1641, [arXiv:0806.4194 \[hep-ph\]](#).
- [16] N. D. Christensen, P. de Aquino, C. Degrande, C. Duhr, B. Fuks, *et al.*, *Eur.Phys.J.* **C71** (2011) 1541, [arXiv:0906.2474 \[hep-ph\]](#).
- [17] F. Staub, *Comput.Phys.Commun.* **181** (2010) 1077–1086, [arXiv:0909.2863 \[hep-ph\]](#).
- [18] A. Alloul, N. D. Christensen, C. Degrande, C. Duhr, and B. Fuks, *Comput.Phys.Commun.* **185** (2014) 2250–2300, [arXiv:1310.1921 \[hep-ph\]](#).
- [19] F. Staub, *Comput.Phys.Commun.* **185** (2014) 1773–1790, [arXiv:1309.7223 \[hep-ph\]](#).
- [20] S. Alioli, P. Nason, C. Oleari, and E. Re, *JHEP* **1006** (2010) 043, [arXiv:1002.2581 \[hep-ph\]](#).
- [21] F. Cascioli, P. Maierhofer, and S. Pozzorini, *Phys.Rev.Lett.* **108** (2012) 111601, [arXiv:1111.5206 \[hep-ph\]](#).
- [22] Z. Bern, L. Dixon, F. Febres Cordero, S. Höche, H. Ita, *et al.*, *J.Phys.Conf.Ser.* **523** (2014) 012051, [arXiv:1310.2808 \[hep-ph\]](#).

- [23] J. Alwall, R. Frederix, S. Frixione, V. Hirschi, F. Maltoni, *et al.*, *JHEP* **1407** (2014) 079, [arXiv:1405.0301 \[hep-ph\]](#).
- [24] J. Kublbeck, M. Bohm, and A. Denner, *Comput.Phys.Commun.* **60** (1990) 165–180.
- [25] T. Hahn, *Comput.Phys.Commun.* **140** (2001) 418–431, [arXiv:hep-ph/0012260 \[hep-ph\]](#).
- [26] LHAPDF, <http://lhapdf.hepforge.org>.
- [27] J. Vermaseren, [arXiv:math-ph/0010025 \[math-ph\]](#).
- [28] M. Jamin and M. E. Lautenbacher, *Comput.Phys.Commun.* **74** (1993) no. 2, 265 – 288.
- [29] T. Hahn, *Nucl.Phys.Proc.Suppl.* **116** (2003) 363–367, [arXiv:hep-ph/0210220 \[hep-ph\]](#).
- [30] T. Hahn, *Nucl.Phys.Proc.Suppl.* **135** (2004) 333–337, [arXiv:hep-ph/0406288 \[hep-ph\]](#).
- [31] T. Hahn, *eConf* **C050318** (2005) 0604, [arXiv:hep-ph/0506201 \[hep-ph\]](#).
- [32] T. Hahn and M. Rauch, *Nucl.Phys.Proc.Suppl.* **157** (2006) 236–240, [arXiv:hep-ph/0601248 \[hep-ph\]](#).
- [33] T. Hahn and J. Illana, *Nucl.Phys.Proc.Suppl.* **160** (2006) 101–105, [arXiv:hep-ph/0607049 \[hep-ph\]](#).
- [34] T. Hahn, *Comput.Phys.Commun.* **178** (2008) 217–221, [arXiv:hep-ph/0611273 \[hep-ph\]](#).
- [35] R. Mertig, M. Böhm, and A. Denner, *Comput.Phys.Commun.* **64** (1991) no. 3, 345 – 359.
- [36] D. Ghosh and M. Wiebusch, [arXiv:1411.2029 \[hep-ph\]](#).